

Simuler un transformateur Fonctionnement en non-linéaire

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :
[page d'accueil](#)


d'autres pdf, sur différents sujets :
[liste des PDF](#)

Dans ce document, nous allons présenter la conséquence de la saturation sur un transformateur et comment exploiter Pspice pour faire des simulations correspondantes.

1) Comment modéliser un circuit magnétique

- Rappelons que si le transfo est à air, on a $\boxed{B = \mu_0 H}$, avec :

B : champ magnétique (en T), μ_0 perméabilité du vide (en H/m)
H, excitation magnétique (en A/m ou ampère tour/m).



On a $H = N I / l$, où N : nombre de spires, I intensité (en A), l : longueur du circuit magnétique (en m).

Remarque : dans un transformateur à air, le couplage magnétique est très faible : de nombreuses lignes de champ se referment à l'extérieur des bobinages. Le coefficient K est inférieur à 1.

- Si le transfo est à circuit magnétique de perméabilité μ_r constante, on a : $\boxed{B = \mu_0 \mu_r H}$. On conserve une relation linéaire entre B et H, mais avec une pente μ_r fois plus élevée. (μ_r sans dimension).


Le circuit magnétique canalise les lignes de champ : si les bobinages primaire et secondaire sont serrés entre eux, le couplage est élevé, K est proche de 1.

- Le circuit magnétique du transformateur peut présenter un phénomène de saturation et d'hystérésis. La simulation Pspice doit alors se faire avec un modèle précis de circuit magnétique.

Plus finement, le modèle du circuit magnétique doit également tenir compte de la température.

Les circuits magnétiques utilisés pour les transformateurs sont du type « doux ».

Caractéristiques $B = f(H)$ de circuit magnétique.



On montre qu'à chaque période, une part de l'énergie traversant le transformateur est perdue dans le circuit magnétique. Ces « pertes fer » (en J/m^3) sont proportionnelles à S, la surface du cycle d'hystérésis, et ont pour équation $P_f = V S f$. Les constructeurs de circuits magnétiques donnent des abaques de puissance volumétrique (en kW/m^3), en fonction de B et paramétrées en fréquence.

La conséquence du phénomène de saturation est qu'à tension sinusoïdale, le courant ne l'est pas.

Le courant se décompose donc en fondamental + harmoniques. Les harmoniques ont des amplitudes de plus en plus faibles avec la fréquence, mais les pertes magnétiques augmentent avec la fréquence. Par conséquence, les premières harmoniques contribuent de façon non négligeable sur les pertes fer.

Les pertes par courant de Foucault (courants induits dans la masse magnétique) sont incluses dans les pertes fer (on considère le cycle d'hystérésis dynamique et non statique), mais sont négligeables en haute fréquence car les matériaux magnétiques utilisés sont mauvais conducteurs électriques.

En première approximation, ces pertes fer sont modélisables par une résistance R_f , placée en // sur L_1 .

Parmi les différentes façons de modéliser un circuit magnétique par Pspice, il y a le modèle de **Jiles et Atherton**. Ce modèle inclut le phénomène de saturation et d'hystérésis.

Il existe le modèle de **Tabrizi** (LEVEL=3) mais non utilisable avec la version 17.2 LITE.

Après avoir défini L1, L2, un transformateur avec **un modèle spécifique de noyau magnétique** est défini par Pspice par : `K<name> L1 L2 K Model_core`
 Dans ce cas, L1 L2 représente le nombre de spires des enroulements, et non leur valeur en Henry.
 K est le coefficient de couplage (entre 0 et 1).

Attention : les unités du domaine « magnétisme » utilisées par Pspice ne sont pas dans le Système International. Pspice utilise le Gauss, et l'Oersted. Pour revenir en S.I., il faut exploiter :

Pour le champ magnétique : $1 \text{ T} = 10^4 \text{ Gauss}$, pour l'excitation : $1 \text{ A/m} = 1 \text{ Oersted} \times 4 \pi \cdot 10^{-3}$

1) Il existe principalement 3 formes physiques de circuits magnétiques : pot, carré, tore. Quelle que soit la forme, les lignes de champ traversent une section et suivent une longueur qui sont définies par :

AREA : section (en **cm²**) PATH : longueur (en **cm**)

2) Il existe une multitude de matériaux magnétiques, à choisir selon la fréquence, l'intensité du champ, la tenue en température, etc. : le fer, certains aciers de fer et nickel, des ferrites ...

Quelle que soit la matière, le matériau magnétique est spécifié par (modèle de **Jiles et Atherton**) :

MS : aimantation à saturation (en **Gauss**.)

A : paramètre d'énergie thermique (**A/m**)

C : facteur de courbure

K : paramètre d'anisotropie (indique indirectement la surface du cycle d'hystérésis)

On peut ajouter : GAP : longueur de l'entrefer (**cm**).

PACK : coefficient de foisonnement pour les circuits feuilletés.

Des exemples de constructeurs : Ferroxcube, Siemens Matsushita, Philips, TDK, Tokin, Amidon...

L'écriture de la netlist Pspice peut être faite de 2 façons :

façon 1 : On passe tous les paramètres (matériau, dimensions) dans une seule ligne. Par exemple :

```
L1 3 8 15 ; bobinage primaire entre les noeuds 3 et 8 avec 15 spires
L2 4 6 45 ; bobinage secondaire entre les noeuds 4 et 6 avec 45 spires
K2 L1 L2 0.999 Transfo ; couplage non parfait, et modèle donné.
```

Les paramètres et les dimensions du tore magnétique sont donnés par la primitive "CORE":

```
.MODEL Transfo CORE (MS=415.2K A=44.82 C=.4112 K=25.74 AREA=1.17 PATH=8.49)
```

façon 2 : Si on simule plusieurs transformateurs ayant tous le même matériau (ici 3B7) :

on définit un modèle pour le matériau, et on passe les dimensions sur une autre ligne :

```
L1 1 0 10 ; 10 spires entre 1 et 0
L2 2 0 100 ; 100 spires entre 2 et 0
Ktransfo L1 L2 0.995 transfo1
```


La netlist doit faire apparaître alors 2 lignes :

```
.MODEL 3B7 CORE (MS=342.012K A=15.0199 K=712.814M ) ; matériau pour les transfos
.MODEL transfo1 AKO:3B7 CORE (AREA=0.970 PATH=4.61) ; dimension du transfo1
```

Dans la syntaxe Pspice, AKO signifie A KIND OF.

L'entrefer (en cm) doit être précisé avec le matériau, (et non avec la dimension).

- On peut créer, dans le circuit magnétique, un entrefer (GAP). C'est un espace sans circuit magnétique dans le chemin des lignes de champ. Une fine cale d'un matériau de perméabilité $\mu_r \approx 1$ est utilisée. La conséquence de cette mise en série des réluctances est de créer un matériau équivalent de perméabilité globale plus faible. A même H, le champ B sera réduit. La conséquence recherchée est de travailler hors saturation d'une part, et d'être moins tributaire des propriétés magnétiques du matériau (sujettes à fluctuation) d'autre part.



Influence de l'entrefer.

Quelques exemples de modèles Pspice de circuits magnétiques :

Consulter, dans le répertoire d'installation,SPB_17.2\tools\pspice\library\core.lib et magnetic.lib

Matériaux simulables par le modèle de Jiles et Atherton disponibles par l'installation de Pspice Lite :

2P40, 2P50, 2P65, 2P80, 2P90,	3B1, 3B7, 3B8, 3B46	3C11, 3C20, 3C34, 3C81, 3C85, 3C90, 3C91, 3C92, 3C93, 3C94, 3C95, 3C96,	3D3,
3E5, 3E6, 3E7, 3E8, 3E9, 3E25, 3E26, 3E27, 3E28, 3E55,	3F3, 3F4, 3F5, 3F35, 3F45,	3H1,	3S4,
4A11, 4A15, 4A20,	4B1, 4B2, 4B3	4C6, 4C65,	4D2, 4F1.
K3E5, K3E2A, K3D3, K3C8, K3C6A,	K3B9, K3B7.		

Des **formes** standards de circuits magnétiques pour transformateurs :

- Forme en « E », (repérée par les lettres E, EC, ETD, EFD...),
- Pots (repérés par les lettres RM), dont les bobines sont circulaires,
- Forme en « U »,
- Forme toroïdale.

2) Simulation d'une unique self bobinée sur un circuit magnétique

La présence d'un circuit magnétique affecte d'un coefficient μ_r la perméabilité.

L'équation donnant la valeur de l'inductance est $L = \mu_0 \mu_r N^2 S / l$ (en système international). Rappelons que cette équation est applicable si toutes les lignes de champ se referment sur elles-mêmes.

On exploite un tore TX/22/14/6.4 à base d'un matériau magnétique 3D3 (modèle Pspice dans la librairie).

Des extraits de la datasheet du fabricant Ferroxcube sont donnés :

3D3 SPECIFICATIONS


A medium frequency filter and tuning material optimized for frequencies from 0.2 up to 2 MHz.

Effective core parameters

SYMBOL	PARAMETER	VALUE	UNIT
$\Sigma(l/A)$	core factor (C1)	2.20	mm ⁻¹
V_e	effective volume	1340	mm ³
l_e	effective length	54.2	mm
A_e	effective area	24.8	mm ²
m	mass of core	≈ 6.5	g

GRADE	A_L (nH)	μ_i	TYPE NUMBER
3D3	454 ± 20%	≈ 750	TX22/14/6.4-3D3

TX22/14/6.4 Ferrite toroids



Observons les dimensions du tore :
 Epaisseur $6,75 - 2 \times 0,12 = 6,51$ mm
 Diamètre extérieur : $22,35 - (2 \times 0,12) = 22,11$ mm
 Diamètre intérieur : $13,47 + (2 \times 0,12) = 13,71$ mm

La valeur 54, 2 mm pour la longueur effective se justifie par le calcul simple :

Diamètre moy = $(22,11 \text{ mm} + 13,71 \text{ mm}) / 2 = 17,91 \text{ mm}$

D'où un périmètre moyen : $\pi \times 17,91 \text{ mm} = 56,26 \text{ mm}$

La valeur 24,8 mm² pour l'aire effective se justifie par le calcul simple :

Section, supposée rectangulaire = $(22,11 - 13,71) / 2 \times 6,51 = 27,3 \text{ mm}^2$.

C'est un peu moins à cause des coins arrondis.

On prend un bobinage composé d'une couche de $N = 100$ spires, de diamètre 0,41 mm (émaillé, jauge 26). Ces 100 spires sont jointives au centre du tore : elles occupent 41 mm, ce qui entre dans le trou du tore de circonférence intérieure 42 mm.

On peut déterminer, par calcul, l'inductance de cet enroulement :


On a, en S.I. : $S = 24,8 \cdot 10^{-6}$, $l = 54,2 \cdot 10^{-3}$, $\mu_0 = 4\pi \cdot 10^{-7}$, $\mu_r = \mu_i$ (intrinsèque) = 750 (à 25 °).
 Il vient : $L = \mu_0 \mu_r N^2 S / l = 4\pi \cdot 10^{-7} \times 750 \times (100)^2 \times 24,8 \cdot 10^{-6} / 54,2 \cdot 10^{-3} = 4,31 \text{ mH}$

Remarque : La data sheet indique $AL = 454 \pm 20 \%$ soit dans [363 ; 544]
 Or $AL = \mu_0 \mu_r S / l = 4 \pi \cdot 10^{-7} \times 750 \times 24,8 \cdot 10^{-6} / 54,2 \cdot 10^{-3} = 4,31 \cdot 10^{-7} \text{ H/N}^2 = 431 \text{ nH/N}^2$, valeur donnée par la datasheet à 5% près. La valeur de AL permet de calculer rapidement l'inductance :
 Une spire donne 431 nH. 100 spires donnent 10 000 fois plus, donc 4,31 mH.

Simulons un circuit RL série, avec $R = 1 \text{ k}\Omega$. La constante de temps attendue est $\tau = L/R = 4,31 \mu\text{s}$.
 D'où la netlist :

Notons :
 PATH = 54,2 mm = 5,42 cm
 AREA = 24,8 mm² = 0,248 cm²
 On exploite un transformateur, dont le secondaire n'est pas connecté (ici relié à une résistance R_L quasi infinie) :

Le schéma est :



```
test d'un bobinage à circuit magnetique
* fichier tore_3D3.cir
* circuit :
Vin 10 0 sin (0 0.15 10k 0 0 0) AC=1
R 10 1 1k
L 1 0 100 ; 100 spires
L2 2 0 1 ; 1 spire
Ktransfo L L2 1 Transfo
RL 2 0 100Meg ; à vide
.MODEL Transfo CORE (Ms=356.1K A=54.37 C=.1291 K=80.59
+ AREA=0.248 PATH=5.42) ;tore TX/22/14/6.4
.TRAN 10n 10m 9.9m 10n ; pour run 10 kHz
*.AC DEC 100 1Meg 10GHz
.probe
.end
```

Netlist prête à simuler

a) Simulation en linéaire de la caractéristique magnétique du matériau 3D3

On vérifie dans un premier temps la perméabilité intrinsèque du 3D3.

En « petits signaux », c'est-à-dire avec un champ magnétique de 0,25 mT, et en « basses fréquences », c'est-à-dire inférieur à 10 kHz, le matériau est donné, par Ferroxcube, à $\mu_r = 750$:

SYMBOL	CONDITIONS	VALUE	UNIT
μ_i	25 °C; ≤10 kHz; 0,25 mT	750 ±20%	

Dans notre simulation, l'amplitude de 0,15 V est choisie pour être à $B = 0,25 \text{ mT}$.

La fréquence est 10 kHz pour être calé sur la datasheet.

D'où la perméabilité :


Trace Name	Y1	Y2	Y1 - Y2
X Values	200.507m	-200.472m	400.979m
B(Ktransfo)/10000	189.291u	-188.991u	378.282u

$$\Delta B / \Delta H = \frac{378,282 \text{ u}}{400,979 \text{ m}} = 0,9434 \text{ m}$$

$$= \mu_0 \mu_r.$$

$$\text{Il vient } \mu_r = \frac{0,9434 \text{ m}}{4 \pi \cdot 10^{-7}} = 750,$$

valeur conforme à la data sheet.



Remarque :

$\mu_0 \mu_r$ est la pente de $B = f(H)$. En petits signaux, on peut afficher directement la valeur de μ_r en visualisant $D(B(Ktransfo)/(10000*4*pi*1e-7))$.


b) Simulation électrique de l'inductance

On reste dans le domaine linéaire, par la directive .AC. La réponse harmonique permet de retrouver la valeur de L, en visualisant V(1), potentiel issu du pont diviseur L, R série dont l'équation est :

$$\frac{jL\omega}{R + jL\omega} = \frac{j(L/R)\omega}{1 + j(L/R)\omega}$$

Connaissant R on peut déduire L.

On a $\tau = L/R$.



La coupure à -3 dB, que l'on pose à f_0 , est au passage à -45° : 36,912 kHz.

Or $f_0 = 1 / (2 \pi \tau)$. D'où :

$$L = \frac{R}{2\pi f_0} = \frac{1000}{2\pi \cdot 36,912 \cdot 10^3} = 4,31 \text{ mH, ce qui est la valeur prédéterminée.}$$

X Values	36.912K
P(V(1))	45.000
DB(V(1))	-3.0103

c) Matériau 3D3 en non linéaire (saturation, hystérésis)

Vérifions maintenant le comportement en non linéaire. Pour ce faire, il faut travailler à fort H, c'est-à-dire à fort courant.

On peut choisir de réduire la résistance, ou d'augmenter la tension, ou de réduire la fréquence, ou un mix des 3.


Nous avons choisi de placer une sinusoïde de 100 Hz, d'amplitude 2,5 V, et de limiter le courant par une résistance série de 10 Ω .

```
Vin 10 0 sin (0 2.5 100)
R 10 1 10
.TRAN 1u 100m 50m 1u
.end
```

3 lignes à modifier

Dans ces conditions, le courant est très distordu.

C'est la conséquence de travailler en fort signal, car on balaie tout le cycle d'hystérésis.




On peut vérifier ce point par un tracé de la caractéristique $B = f(H)$ de ce même run :

On peut lire :

Champ coercitif : ≈ 50 A/m

Rémanence : ≈ 200 mT.



3) Simulation d'un transformateur

On reprend le tore TX/22/14/6.4 en 3D3.

Le primaire est constitué de 20 spires : il présente une inductance $L_1 = 431 \text{ nH} \times 20^2 = 172,4 \text{ }\mu\text{H}$.

Le secondaire de 40 spires : $L_2 = 431 \text{ nH} \times 40^2 = 689,6 \text{ }\mu\text{H}$.

Par simplicité, on utilise le même fil pour les 2 bobinages.

Si on veut que les (20 + 40) spires tiennent sur une seule couche, il faut un fil de diamètre 42 mm / 60 = 0,7 mm, soit une jauge SWG 22, qui a une section $s = 0,385 \text{ mm}^2$.

Une spire présente une longueur de $(2 \times 4,44 \text{ mm}) + (2 \times 6,75) = 22,38 \text{ mm}$.

Le primaire sera formé d'une longueur de fil de $20 \times 22,38 \approx 4,5 \text{ m}$,

Par application de $R = \rho \frac{l}{s}$, il présentera une résistance de $1,710^{-8} \frac{4,5}{0,385 \cdot 10^{-6}} = 0,2 \text{ }\Omega$.

Egalement, le secondaire sera formé de 9 m de fil, qui présentera $0,4 \text{ }\Omega$.

Il s'agit des valeurs « en continu », donc sans effet de peau, et à température ambiante.

Soit un couplage $K = 0,99$. D'où l'inductance de fuite : $L_f = (1 - K^2) L_2 = 13,73 \text{ }\mu\text{H}$.

Soit une charge formée d'une résistance $R_L = 47 \text{ }\Omega$.

Il a été établi, sous [..\TRANSTFO_L\transfo_lineaire.pdf](#) que l'utilisation d'un transformateur devait se faire dans une bande de fréquence délimitée par :

Equations	Application numérique
$f \gg \frac{R_L}{2\pi L_2}$ pour $\vec{I}_1 = -m\vec{I}_2$	$f \gg \frac{47}{2\pi 690 \mu} = 10,84 \text{ kHz}$
$\frac{r_1}{2\pi L_1} \ll f \ll \frac{R_L}{2\pi L_f}$ pour $\vec{V}_2 = -m\vec{V}_1$	$\frac{0,2}{2\pi 172,4 \mu} = 184,6 \text{ Hz} \ll f \ll \frac{47}{2\pi 13,73 \mu} = 0,545 \text{ MHz}$

Le schéma d'étude est :




Schéma simulé

Sur un transformateur parfait, avec 40 spires et 20 spires, le rapport de transformation est de 2.

Ici, on attend des chutes de tension dans r_p , r_s d'une part, et due à la réponse harmonique du transformateur d'autre part.

Nous allons d'abord vérifier les bandes de fréquences d'emploi de ce transformateur.

Transformateur torique
* fichier transfo_3D3.cir
* circuit :


```
Vin 10 0 sin (0 20 100k) AC 1
rp 10 1 0.2 ; primaire
L1 1 0 20 ; 20 spires
L2 0 2 40 ; 40 spires
Ktransfo L1 L2 0.99 Transfo
rs 2 20 0.4 ; secondaire
RL 20 0 47 ; charge
```

```
.MODEL Transfo CORE (Ms=356.1K A=54.37
+ C=.1291 K=80.59
+ AREA=0.248 PATH=5.42) ; TX/22/14/6.4

.AC DEC 100 1 100MEG
.TRAN 100n 100m 99.95m 100n
.probe
.end
```

Netlist prête à simuler

a) Réponse harmonique



Réponse harmonique du transformateur débitant sur R_L
De haut en bas : tension, courant, puissance

Measurement	Value
Cutoff_Highpass_3dB(V(20))	185.57586
Cutoff_Lowpass_3dB(V(20))	568.96542k
Cutoff_Highpass_3dB(I(L2)/I(L1))	11.18456k

La réponse **en tension** du transformateur extrinsèque est $V(20)/V(10)$. La bande passante en tension est [186 Hz ; 569 kHz]. Dans cette bande passante, on a une atténuation créée par les résistances internes. C'est la raison pour laquelle on n'a pas la valeur 2, mais légèrement en dessous.

La réponse **en courant** du transformateur est $I(L2)/I(L1)$. Cette réponse est du type passe haut avec une fréquence de coupure de 11,2 kHz. Dans le plateau de cette réponse, on retrouve la valeur $1/m = 0,5$.

La réponse harmonique **en puissance** est donc réduite : la résultante des transferts de tension ET courant passe par un maximum (proche de 1) vers 100 kHz.


Remarque importante : la simulation .AC linéarise le circuit étudié, et travaille en petits signaux. La caractéristique magnétique est réduite à l'expression $B = \mu_0 \mu_r H$, avec $\mu_0 \mu_r = 4 \pi 10^{-7} \times 750 = 9,42 10^{-4}$. On ne peut donc pas voir dans cette analyse le comportement en fréquence du circuit magnétique.

b) Réponse temporelle

On choisit d'injecter en entrée un signal sinusoïdal, de fréquence 100 kHz, c'est-à-dire dans la bande passante en puissance.

Il faut un run *très* long pour être en régime permanent (attendre 100 ms avec un signal de 100 kHz, revient à laisser passer 10 000 périodes). Le régime permanent est atteint quand le cycle d'hystérésis est bien centré sur l'origine. La directive .TRAN montre qu'on enregistre qu'à partir de 99,95 ms, pour éviter un fichier .dat trop volumineux.

Le choix de 20 V d'amplitude permet de travailler en « forts signaux », c'est-à-dire ici avec un large cycle d'hystérésis. La valeur efficace est donc 14,14 V.




On visualise la tension secondaire, aux bornes de R_L : son amplitude est 38,96 V, légèrement en dessous de 2×20 V, comme prédéterminé en AC.

Le courant primaire a pour amplitude 1,7175 A. Le déphasage est quasi nul (la contribution de $L_f \omega / m^2 = 2,16 \Omega$ est faible, et l'opposition de phase visualisée est liée à notre convention de signe). L'impédance vue du primaire est donc équivalente à une résistance égale à $20/1,7175 = 11,6448 \Omega$. Avec un transformateur parfait (point de vue courant et circuit magnétique), on aurait, par le théorème de transfert d'impédance, $(47+0,4)/2^2 + 0,2 = 12,05 \Omega$.

Sur ce même run, on peut afficher le cycle d'hystérésis, en S.I. Il est clair qu'on ne travaille plus en petits signaux, car le cycle est décrit largement.

Il est intéressant de dresser le **bilan des puissances** :



Les puissances sont données par la valeur moyenne de $p(t)$. On choisit d'intégrer sur 4 périodes.

X Values	99.995m
-AVGX(W(in),40u)	17.237
AVGX(W(rp),40u)	297.769m
AVGX(W(L1),40u)	16.940
-AVGX(W(L2),40u)	16.342
AVGX(W(rs),40u)	137.908m
AVGX(W(RL),40u)	16.204

Mesuré en fin de calcul de valeur moyenne :

- La source fournit 17,237 W.
- La résistance primaire consomme 0,297 W.
- L'inductance primaire reçoit 16,940 W.
- Le circuit magnétique transfère 16,342 W
- La résistance secondaire consomme 0,138 W.
- La charge consomme 16,204 W

Explications
 $\approx U^2/R = 200 / 11,6448 = 17,175$ W
 On retrouve 17,237 W – 0,297 W
Perte de 16,94 – 16,34 = 0,6 W
 On retrouve 16,342 W – 0,138 W

Le bilan complet montre :

- que les résistances internes dissipent peu (0,297 W et 0,138 W)
- que le rendement global est très bon : $16,204 / 17,237 = 94 \%$.
- **que le circuit magnétique dissipe 0,6 W.**

Examinons plus finement ce dernier point :

La puissance perdue dans le circuit magnétique (appelée pertes fer) est donnée par la relation $P_f = V S f$.

V : volume du circuit magnétique.

Il peut être calculé par : section x longueur moyenne, soit $24,8 \text{ mm}^2 \times 56,26 \text{ mm} = 1395 \text{ mm}^3$

S : surface du cycle d'hystérésis.

Elle peut être grossièrement estimée en comptant les cases rectangulaires inscrites dans le cycle.

On a alors 44 u.a. (1 u.a. = 1 unité d'aire = $2 \text{ A/m} \times 50 \text{ mT} = 0,1$).

La surface du cycle d'hystérésis est : $S = 44 \times 0,1 = 4,4 \text{ J/m}^3$.

f : fréquence = 100 kHz.

Par $P_f = V S f$, on a $1395 \cdot 10^{-9} \times 4,4 \times 100 \text{ k} \approx 0,6 \text{ W}$.

Ce calcul élémentaire est confirmé par l'outil Pspice calculator « hysteresis core loss ».

Il faut placer un écart de temps qui doit être au moins une période du signal, pour que le cycle soit parcouru entièrement.

Il donne la surface du cycle : 556,47 en Gauss.Oersted soit, en S.I. :

$556,47 / (10^4 \times 4 \pi \cdot 10^{-3}) = 4,428 \text{ J/m}^3$

En plaçant 1395 dans le champ Core Volume, le résultat est 0,617 W.

Core	Ktransfo	Core Volume (mm ³)	1395
Start Time (Seconds)	9.99e-002	End Time (Seconds)	1.00e-001
Frequency (Hz)	99995.33	Core Loss (Watts)	6.17e-001
Hysteresis loop area = 556.47		<input type="button" value="Calculate"/> <input type="button" value="Reset"/> <input type="button" value="Close"/>	

Conclusion

Ce document a montré l'utilisation de Pspice avec un « vrai » transformateur, c'est-à-dire connu par son circuit magnétique (matériau et forme) et le nombre de spires primaire et secondaire. On retrouve des résultats en accord avec la théorie (transformateur parfait et ses défauts), et qui tient compte du comportement du matériau magnétique.

Néanmoins, dans ce document qui se voulait facile à comprendre, on n'a pas tenu compte de l'effet de peau (phénomène important à 100 kHz), ni de la température (μ_r est sensible à la température) ...

De plus, une fluctuation, même minime, du coefficient de couplage change beaucoup la valeur de l'inductance de fuite, et donc modifie la bande passante du transformateur (en tension et en puissance).

Une simulation qui se voudrait réaliste d'un transformateur vrai, notamment en hautes fréquences, est donc un travail bien plus délicat.

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphasé sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

90	Un anémomètre à fil chaud : simulation comportementale	présentation	document PDF
91	Un oscillateur à pont de Wien stabilisé par CTN	présentation	document PDF
92	L'emballement thermique d'une diode	présentation	document PDF
93	Les puissances dans un amplificateur	présentation	document PDF
94	Asservissement de puissance dans une résistance	présentation	document PDF
95	Asservissement de la puissance émise par une antenne radio	présentation	document PDF
96	Un driver de LED de puissance	présentation	document PDF
97	Exploiter Pspice pour simuler des filtres numériques	présentation	document PDF
98	Un filtre en cosinus surélevé avec Pspice	présentation	document PDF
99	Effet de la température sur un amplificateur en classe A	présentation	document PDF
100	Un amplificateur à transistors JFET et bipolaires	présentation	document PDF
Supplément, hors article :			
mon cours « Electronique pour les communications numériques », polycopié couleur 201 pages en pdf			

[retour à l'écran d'accueil de ce site](#)